Rules for integrands involving inert trig functions

0. $\int (a\,F[c+d\,x]^p)^n\,dx$ when $F\in\{\mathtt{Sin},\,\mathtt{Cos},\,\mathtt{Tan},\,\mathtt{Cot},\,\mathtt{Sec},\,\mathtt{Csc}\}\,\land\,n\notin\mathbb{Z}\,\land\,p\in\mathbb{Z}$

   1: $\int (a\,F[c+d\,x]^p)^n\,dx$ when $F\in\{\mathtt{Sin},\,\mathtt{Cos},\,\mathtt{Tan},\,\mathtt{Cot},\,\mathtt{Sec},\,\mathtt{Csc}\}\,\land\,n\notin\mathbb{Z}\,\land\,p\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a\,F[c+d\,x]^p)^n}{F[c+d\,x]^{n\,p}} == 0$

Rule: If $F\in\{\mathtt{Sin},\,\mathtt{Cos},\,\mathtt{Tan},\,\mathtt{Cot},\,\mathtt{Sec},\,\mathtt{Csc}\}\,\land\,n\notin\mathbb{Z}\,\land\,p\in\mathbb{Z}$, then

$$\int (a\,F[c+d\,x]^p)^n\,dx\;\longrightarrow\;\frac{(a\,F[c+d\,x]^p)^n}{F[c+d\,x]^{n\,p}}\int F[c+d\,x]^{n\,p}\,dx$$

Program code:

```
Int[(a_.*F_[c_.+d_.*x_]^p_)^n_,x_Symbol] :=
  With[{v=ActivateTrig[F[c+d*x]]},
  a^IntPart[n]*(v/NonfreeFactors[v,x])^(p*IntPart[n])*(a*v^p)^FracPart[n]/NonfreeFactors[v,x]^(p*FracPart[n])*
    Int[NonfreeFactors[v,x]^(n*p),x]] /;
FreeQ[{a,c,d,n,p},x] && InertTrigQ[F] && Not[IntegerQ[n]] && IntegerQ[p]
```

2: $\int \left(a\,(b\,F[c+d\,x])^p\right)^n\,dx$ when $F \in \{\text{Sin, Cos, Tan, Cot, Sec, Csc}\} \land n \notin \mathbb{Z} \land p \notin \mathbb{Z}$

- Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{(a\,(b\,F[c+d\,x])^p)^n}{(b\,F[c+d\,x])^{n\,p}} == 0$

- Rule: If $F \in \{\text{Sin, Cos, Tan, Cot, Sec, Csc}\} \land n \notin \mathbb{Z} \land p \notin \mathbb{Z}$, then

$$\int \left(a\,(b\,F[c+d\,x])^p\right)^n\,dx \longrightarrow \frac{a^{\text{IntPart}[n]}\,\left(a\,(b\,F[c+d\,x])^p\right)^{\text{FracPart}[n]}}{(b\,F[c+d\,x])^{p\,\text{FracPart}[n]}} \int (b\,F[c+d\,x])^{n\,p}\,dx$$

- Program code:

```
Int[(a_.*(b_.*F_[c_.+d_.*x_])^p_)^n_.,x_Symbol] :=
  With[{v=ActivateTrig[F[c+d*x]]},
  a^IntPart[n]*(a*(b*v)^p)^FracPart[n]/(b*v)^(p*FracPart[n])*Int[(b*v)^(n*p),x]] /;
FreeQ[{a,b,c,d,n,p},x] && InertTrigQ[F] && Not[IntegerQ[n]] && Not[IntegerQ[p]]
```

1. $\int F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Trig}[a+b\,x]\,dx$

   1: $\int F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Cos}[a+b\,x]\,dx$

Reference: G&R 2.503, CRC 483

Reference: G&R 2.502, CRC 482

Derivation: Integration by substitution

Basis: $F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Cos}[a+b\,x] == \frac{1}{b}\,F[\mathrm{Sin}[a+b\,x]]\,\partial_x\mathrm{Sin}[a+b\,x]$

Rule:

$$\int F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Cos}[a+b\,x]\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\int F[x]\,dx,\,x,\,\mathrm{Sin}[a+b\,x]\right]$$

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sin[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Sin[c*(a+b*x)]/d,u,x],x],x,Sin[c*(a+b*x)]/d] /;
 FunctionOfQ[Sin[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Cos] || EqQ[F,cos])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cos[c*(a+b*x)],x]},
  -d/(b*c)*Subst[Int[SubstFor[1,Cos[c*(a+b*x)]/d,u,x],x],x,Cos[c*(a+b*x)]/d] /;
 FunctionOfQ[Cos[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Sin] || EqQ[F,sin])
```

```
Int[u_*Cosh[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sinh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Sinh[c*(a+b*x)]/d,u,x],x],x,Sinh[c*(a+b*x)]/d] /;
 FunctionOfQ[Sinh[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x]
```

```
Int[u_*Sinh[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cosh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Cosh[c*(a+b*x)]/d,u,x],x],x,Cosh[c*(a+b*x)]/d] /;
 FunctionOfQ[Cosh[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x]
```

2: $\int F[\sin[a+b x]] \cot[a+b x] \, dx$

Reference: G&R 2.503, CRC 483

Reference: G&R 2.502, CRC 482

Derivation: Integration by substitution

Basis: $F[\sin[a+b x]] \cot[a+b x] == \frac{F[\sin[a+b x]]}{b \sin[a+b x]} \partial_x \sin[a+b x]$

Rule:

$$\int F[\sin[a+b x]] \cot[a+b x] \, dx \to \frac{1}{b} \text{Subst}\left[\int \frac{F[x]}{x} \, dx, x, \sin[a+b x]\right]$$

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sin[c*(a+b*x)],x]},
  1/(b*c)*Subst[Int[SubstFor[1/x,Sin[c*(a+b*x)]/d,u,x],x],x,Sin[c*(a+b*x)]/d] /;
 FunctionOfQ[Sin[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Cot] || EqQ[F,cot])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cos[c*(a+b*x)],x]},
  -1/(b*c)*Subst[Int[SubstFor[1/x,Cos[c*(a+b*x)]/d,u,x],x],x,Cos[c*(a+b*x)]/d] /;
 FunctionOfQ[Cos[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Tan] || EqQ[F,tan])
```

```
Int[u_*Coth[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sinh[c*(a+b*x)],x]},
  1/(b*c)*Subst[Int[SubstFor[1/x,Sinh[c*(a+b*x)]/d,u,x],x],x,Sinh[c*(a+b*x)]/d] /;
 FunctionOfQ[Sinh[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x]
```

```
Int[u_*Tanh[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cosh[c*(a+b*x)],x]},
  1/(b*c)*Subst[Int[SubstFor[1/x,Cosh[c*(a+b*x)]/d,u,x],x],x,Cosh[c*(a+b*x)]/d] /;
 FunctionOfQ[Cosh[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x]
```

2. $\int F[\mathrm{Tan}[a+b\,x]]\,\mathrm{Trig}[a+b\,x]^n\,dx$

1: $\int F[\mathrm{Tan}[a+b\,x]]\,\mathrm{Sec}[a+b\,x]^2\,dx$

Reference: G&R 2.504

Derivation: Integration by substitution

Basis: $F[\mathrm{Tan}[a+b\,x]]\,\mathrm{Sec}[a+b\,x]^2 == \frac{1}{b}\,F[\mathrm{Tan}[a+b\,x]]\,\partial_x \mathrm{Tan}[a+b\,x]$

Rule:

$$\int F[\mathrm{Tan}[a+b\,x]]\,\mathrm{Sec}[a+b\,x]^2\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\int F[x]\,dx,\,x,\,\mathrm{Tan}[a+b\,x]\right]$$

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  With[{d=FreeFactors[Tan[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Tan[c*(a+b*x)]/d,u,x],x],x,Tan[c*(a+b*x)]/d] /;
 FunctionOfQ[Tan[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && NonsumQ[u] && (EqQ[F,Sec] || EqQ[F,sec])
```

```
Int[u_/cos[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  With[{d=FreeFactors[Tan[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Tan[c*(a+b*x)]/d,u,x],x],x,Tan[c*(a+b*x)]/d] /;
 FunctionOfQ[Tan[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && NonsumQ[u]
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  With[{d=FreeFactors[Cot[c*(a+b*x)],x]},
  -d/(b*c)*Subst[Int[SubstFor[1,Cot[c*(a+b*x)]/d,u,x],x],x,Cot[c*(a+b*x)]/d] /;
 FunctionOfQ[Cot[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && NonsumQ[u] && (EqQ[F,Csc] || EqQ[F,csc])
```

```
Int[u_/sin[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  With[{d=FreeFactors[Cot[c*(a+b*x)],x]},
  -d/(b*c)*Subst[Int[SubstFor[1,Cot[c*(a+b*x)]/d,u,x],x],x,Cot[c*(a+b*x)]/d] /;
 FunctionOfQ[Cot[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && NonsumQ[u]
```

```
Int[u_*Sech[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  With[{d=FreeFactors[Tanh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Tanh[c*(a+b*x)]/d,u,x],x],x,Tanh[c*(a+b*x)]/d] /;
 FunctionOfQ[Tanh[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && NonsumQ[u]
```

```
Int[u_*Csch[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  With[{d=FreeFactors[Coth[c*(a+b*x)],x]},
  -d/(b*c)*Subst[Int[SubstFor[1,Coth[c*(a+b*x)]/d,u,x],x],x,Coth[c*(a+b*x)]/d] /;
 FunctionOfQ[Coth[c*(a+b*x)]/d,u,x,True]] /;
FreeQ[{a,b,c},x] && NonsumQ[u]
```

2: $\int F[\mathrm{Tan}[a+b\,x]]\,\mathrm{Cot}[a+b\,x]^n\,dx$ when $n\in\mathbb{Z}$

Reference: G&R 2.504

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z}$, then $F[\operatorname{Tan}[a+b x]] \operatorname{Cot}[a+b x]^n == \frac{F[\operatorname{Tan}[a+b x]]}{b \operatorname{Tan}[a+b x]^n \left(1+\operatorname{Tan}[a+b x]^2\right)} \partial_x \operatorname{Tan}[a+b x]$

Rule: If $n \in \mathbb{Z}$, then

$$\int F[\operatorname{Tan}[a+b x]] \operatorname{Cot}[a+b x]^n \, dx \longrightarrow \frac{1}{b} \operatorname{Subst}\left[\int \frac{F[x]}{x^n \left(1+x^2\right)} \, dx, x, \operatorname{Tan}[a+b x]\right]$$

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_.,x_Symbol] :=
  With[{d=FreeFactors[Tan[c*(a+b*x)],x]},
  1/(b*c*d^(n-1))*Subst[Int[SubstFor[1/(x^n*(1+d^2*x^2)),Tan[c*(a+b*x)]/d,u,x],x],x,Tan[c*(a+b*x)]/d] /;
 FunctionOfQ[Tan[c*(a+b*x)]/d,u,x,True] && TryPureTanSubst[ActivateTrig[u]*Cot[c*(a+b*x)]^n,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n] && (EqQ[F,Cot] || EqQ[F,cot])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_.,x_Symbol] :=
  With[{d=FreeFactors[Cot[c*(a+b*x)],x]},
  -1/(b*c*d^(n-1))*Subst[Int[SubstFor[1/(x^n*(1+d^2*x^2)),Cot[c*(a+b*x)]/d,u,x],x],x,Cot[c*(a+b*x)]/d] /;
 FunctionOfQ[Cot[c*(a+b*x)]/d,u,x,True] && TryPureTanSubst[ActivateTrig[u]*Tan[c*(a+b*x)]^n,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n] && (EqQ[F,Tan] || EqQ[F,tan])
```

```
Int[u_*Coth[c_.*(a_.+b_.*x_)]^n_.,x_Symbol] :=
  With[{d=FreeFactors[Tanh[c*(a+b*x)],x]},
  1/(b*c*d^(n-1))*Subst[Int[SubstFor[1/(x^n*(1-d^2*x^2)),Tanh[c*(a+b*x)]/d,u,x],x],x,Tanh[c*(a+b*x)]/d] /;
 FunctionOfQ[Tanh[c*(a+b*x)]/d,u,x,True] && TryPureTanSubst[ActivateTrig[u]*Coth[c*(a+b*x)]^n,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n]
```

```
Int[u_*Tanh[c_.*(a_.+b_.*x_)]^n_.,x_Symbol] :=
  With[{d=FreeFactors[Coth[c*(a+b*x)],x]},
  1/(b*c*d^(n-1))*Subst[Int[SubstFor[1/(x^n*(1-d^2*x^2)),Coth[c*(a+b*x)]/d,u,x],x],x,Coth[c*(a+b*x)]/d] /;
 FunctionOfQ[Coth[c*(a+b*x)]/d,u,x,True] && TryPureTanSubst[ActivateTrig[u]*Tanh[c*(a+b*x)]^n,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n]
```

3: $\int F[\tan[a+b\,x]]\,dx$

Reference: G&R 2.504

Derivation: Integration by substitution

Basis: $F[\tan[z]] == \frac{F[\tan[z]]}{1+\tan[z]^2}\,\partial_z \tan[z]$

Rule:

$$\int F[\tan[a+b\,x]]\,dx \to \frac{1}{b}\,\text{Subst}\left[\int \frac{F[x]}{1+x^2}\,dx,\,x,\,\tan[a+b\,x]\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[F[Cot[a+b*x]],x]","-1/b*Subst[Int[F[x]/(1+x^2),x],x,Cot[a+b*x]]",Hold[
  With[{d=FreeFactors[Cot[v],x]},
  Dist[-d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Cot[v]/d,u,x],x],x,Cot[v]/d],x]]]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Cot[v],x],u,x,True] && TryPureTanSubst[ActivateTrig[u],x]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  With[{d=FreeFactors[Cot[v],x]},
  Dist[-d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Cot[v]/d,u,x],x],x,Cot[v]/d],x]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Cot[v],x],u,x,True] && TryPureTanSubst[ActivateTrig[u],x]]]
```

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[F[Tan[a+b*x]],x]","1/b*Subst[Int[F[x]/(1+x^2),x],x,Tan[a+b*x]]",Hold[
  With[{d=FreeFactors[Tan[v],x]},
  Dist[d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v]/d,u,x],x],x,Tan[v]/d],x]]]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Tan[v],x],u,x,True] && TryPureTanSubst[ActivateTrig[u],x]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  With[{d=FreeFactors[Tan[v],x]},
  Dist[d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v]/d,u,x],x],x,Tan[v]/d],x]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Tan[v],x],u,x,True] && TryPureTanSubst[ActivateTrig[u],x]]]
```

3. $\int \text{Trig}[a+b\,x]^p\,\text{Trig}[c+d\,x]^q\cdots\,dx$ when $(p\mid q\mid\cdots)\in\mathbb{Z}^+$

1: $\int \text{Trig}[a+b\,x]^p\,\text{Trig}[c+d\,x]^q\,dx$ when $(p\mid q)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(p\mid q)\in\mathbb{Z}^+$, then

$$\int \text{Trig}[a+b\,x]^p\,\text{Trig}[c+d\,x]^q\,dx\;\longrightarrow\;\int \text{TrigReduce}\left[\text{Trig}[a+b\,x]^p\,\text{Trig}[c+d\,x]^q\right]dx$$

Program code:

```
Int[F_[a_.+b_.*x_]^p_.*G_[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[ActivateTrig[F[a+b*x]^p*G[c+d*x]^q],x],x] /;
FreeQ[{a,b,c,d},x] && (EqQ[F,sin] || EqQ[F,cos]) && (EqQ[G,sin] || EqQ[G,cos]) && IGtQ[p,0] && IGtQ[q,0]
```

2: $\int \mathrm{Trig}[a+b\,x]^p\,\mathrm{Trig}[c+d\,x]^q\,\mathrm{Trig}[e+f\,x]^r\,dx$ when $(p\mid q\mid r)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(p\mid q\mid r)\in\mathbb{Z}^+$, then

$$\int \mathrm{Trig}[a+b\,x]^p\,\mathrm{Trig}[c+d\,x]^q\,\mathrm{Trig}[e+f\,x]^r\,dx \longrightarrow \int \mathrm{TrigReduce}\left[\mathrm{Trig}[a+b\,x]^p\,\mathrm{Trig}[c+d\,x]^q\,\mathrm{Trig}[e+f\,x]^r\right]dx$$

Program code:

```
Int[F_[a_.+b_.*x_]^p_.*G_[c_.+d_.*x_]^q_.*H_[e_.+f_.*x_]^r_.,x_Symbol] :=
  Int[ExpandTrigReduce[ActivateTrig[F[a+b*x]^p*G[c+d*x]^q*H[e+f*x]^r],x],x] /;
FreeQ[{a,b,c,d,e,f},x] && (EqQ[F,sin] || EqQ[F,cos]) && (EqQ[G,sin] || EqQ[G,cos]) && (EqQ[H,sin] || EqQ[H,cos]) && IGtQ[p,0] && IGtQ[q,0] && I
```

4. $\int F[\text{Sin}[a+b\,x]]\,\text{Trig}[a+b\,x]\,dx$

1: $\int F[\text{Sin}[a+b\,x]]\,\text{Cos}[a+b\,x]\,dx$

Reference: G&R 2.503, CRC 483

Reference: G&R 2.502, CRC 482

Derivation: Integration by substitution

Basis: $F[\text{Sin}[a+b\,x]]\,\text{Cos}[a+b\,x] == \frac{1}{b}\,F[\text{Sin}[a+b\,x]]\,\partial_x\text{Sin}[a+b\,x]$

Rule:

$$\int F[\text{Sin}[a+b\,x]]\,\text{Cos}[a+b\,x]\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int F[x]\,dx,\,x,\,\text{Sin}[a+b\,x]\right]$$

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sin[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Sin[c*(a+b*x)]/d,u,x],x],x,Sin[c*(a+b*x)]/d] /;
 FunctionOfQ[Sin[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Cos] || EqQ[F,cos])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cos[c*(a+b*x)],x]},
  -d/(b*c)*Subst[Int[SubstFor[1,Cos[c*(a+b*x)]/d,u,x],x],x,Cos[c*(a+b*x)]/d] /;
 FunctionOfQ[Cos[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Sin] || EqQ[F,sin])
```

```
Int[u_*Cosh[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sinh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Sinh[c*(a+b*x)]/d,u,x],x],x,Sinh[c*(a+b*x)]/d] /;
 FunctionOfQ[Sinh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x]
```

```
Int[u_*Sinh[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cosh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[1,Cosh[c*(a+b*x)]/d,u,x],x],x,Cosh[c*(a+b*x)]/d] /;
 FunctionOfQ[Cosh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x]
```

2: $\int F[\text{Sin}[a+b\,x]]\,\text{Cot}[a+b\,x]\,dx$

Reference: G&R 2.503, CRC 483

Reference: G&R 2.502, CRC 482

Derivation: Integration by substitution

Basis: $F[\text{Sin}[a+b\,x]]\,\text{Cot}[a+b\,x] = \frac{F[\text{Sin}[a+b\,x]]}{b\,\text{Sin}[a+b\,x]}\,\partial_x \text{Sin}[a+b\,x]$

Rule:

$$\int F[\text{Sin}[a+b\,x]]\,\text{Cot}[a+b\,x]\,dx \rightarrow \frac{1}{b}\,\text{Subst}\left[\int \frac{F[x]}{x}\,dx,\ x,\ \text{Sin}[a+b\,x]\right]$$

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sin[c*(a+b*x)],x]},
  1/(b*c)*Subst[Int[SubstFor[1/x,Sin[c*(a+b*x)]/d,u,x],x],x,Sin[c*(a+b*x)]/d] /;
 FunctionOfQ[Sin[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Cot] || EqQ[F,cot])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cos[c*(a+b*x)],x]},
  -1/(b*c)*Subst[Int[SubstFor[1/x,Cos[c*(a+b*x)]/d,u,x],x],x,Cos[c*(a+b*x)]/d] /;
 FunctionOfQ[Cos[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && (EqQ[F,Tan] || EqQ[F,tan])
```

```
Int[u_*Coth[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Sinh[c*(a+b*x)],x]},
  1/(b*c)*Subst[Int[SubstFor[1/x,Sinh[c*(a+b*x)]/d,u,x],x],x,Sinh[c*(a+b*x)]/d] /;
 FunctionOfQ[Sinh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x]
```

```
Int[u_*Tanh[c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{d=FreeFactors[Cosh[c*(a+b*x)],x]},
  1/(b*c)*Subst[Int[SubstFor[1/x,Cosh[c*(a+b*x)]/d,u,x],x],x,Cosh[c*(a+b*x)]/d] /;
 FunctionOfQ[Cosh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x]
```

5. $\int F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Trig}[a+b\,x]^n\,dx$

1: $\int F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Cos}[a+b\,x]^n\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}$

Reference: G&R 2.503, CRC 483

Reference: G&R 2.502, CRC 482

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then $F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Cos}[a+b\,x]^n == \frac{1}{b}\left(1-\mathrm{Sin}[a+b\,x]^2\right)^{\frac{n-1}{2}} F[\mathrm{Sin}[a+b\,x]]\,\partial_x \mathrm{Sin}[a+b\,x]$

–

Rule: If $\frac{n-1}{2}\in\mathbb{Z}$, then

$$\int F[\mathrm{Sin}[a+b\,x]]\,\mathrm{Cos}[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\int \left(1-x^2\right)^{\frac{n-1}{2}} F[x]\,dx,\ x,\ \mathrm{Sin}[a+b\,x]\right]$$

–

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Sin[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[(1-d^2*x^2)^((n-1)/2),Sin[c*(a+b*x)]/d,u,x],x],x,Sin[c*(a+b*x)]/d] /;
 FunctionOfQ[Sin[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Cos] || EqQ[F,cos])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Sin[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[(1-d^2*x^2)^((-n-1)/2),Sin[c*(a+b*x)]/d,u,x],x],x,Sin[c*(a+b*x)]/d] /;
 FunctionOfQ[Sin[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Sec] || EqQ[F,sec])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Cos[c*(a+b*x)],x]},
  -d/(b*c)*Subst[Int[SubstFor[(1-d^2*x^2)^((n-1)/2),Cos[c*(a+b*x)]/d,u,x],x],x,Cos[c*(a+b*x)]/d] /;
 FunctionOfQ[Cos[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Sin] || EqQ[F,sin])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Cos[c*(a+b*x)],x]},
  -d/(b*c)*Subst[Int[SubstFor[(1-d^2*x^2)^((-n-1)/2),Cos[c*(a+b*x)]/d,u,x],x],x,Cos[c*(a+b*x)]/d] /;
 FunctionOfQ[Cos[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Csc] || EqQ[F,csc])
```

```
Int[u_*Cosh[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Sinh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[(1+d^2*x^2)^((n-1)/2),Sinh[c*(a+b*x)]/d,u,x],x],x,Sinh[c*(a+b*x)]/d] /;
 FunctionOfQ[Sinh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u]
```

```
Int[u_*Sech[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Sinh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[(1+d^2*x^2)^((-n-1)/2),Sinh[c*(a+b*x)]/d,u,x],x],x,Sinh[c*(a+b*x)]/d] /;
 FunctionOfQ[Sinh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u]
```

```
Int[u_*Sinh[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Cosh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[(-1+d^2*x^2)^((n-1)/2),Cosh[c*(a+b*x)]/d,u,x],x],x,Cosh[c*(a+b*x)]/d] /;
 FunctionOfQ[Cosh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u]
```

```
Int[u_*Csch[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Cosh[c*(a+b*x)],x]},
  d/(b*c)*Subst[Int[SubstFor[(-1+d^2*x^2)^((-n-1)/2),Cosh[c*(a+b*x)]/d,u,x],x],x,Cosh[c*(a+b*x)]/d] /;
 FunctionOfQ[Cosh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u]
```

2: $\int F[\text{Sin}[a+b\,x]]\,\text{Cot}[a+b\,x]^n\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}$

Reference: G&R 2.503, CRC 483

Reference: G&R 2.502, CRC 482

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then $F[\text{Sin}[a+b\,x]]\,\text{Cot}[a+b\,x]^n == \frac{1}{b}\left(1-\text{Sin}[a+b\,x]^2\right)^{\frac{n-1}{2}}\frac{F[\text{Sin}[a+b\,x]]}{\text{Sin}[a+b\,x]^n}\,\partial_x\text{Sin}[a+b\,x]$

Rule: If $\frac{n-1}{2}\in\mathbb{Z}$, then

$$\int F[\text{Sin}[a+b\,x]]\,\text{Cot}[a+b\,x]^n\,dx\to\frac{1}{b}\,\text{Subst}\left[\int\frac{\left(1-x^2\right)^{\frac{n-1}{2}}F[x]}{x^n}\,dx,\,x,\,\text{Sin}[a+b\,x]\right]$$

Program code:

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Sin[c*(a+b*x)],x]},
  1/(b*c*d^(n-1))*Subst[Int[SubstFor[(1-d^2*x^2)^((n-1)/2)/x^n,Sin[c*(a+b*x)]/d,u,x],x],x,Sin[c*(a+b*x)]/d] /;
 FunctionOfQ[Sin[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Cot] || EqQ[F,cot])
```

```
Int[u_*F_[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Cos[c*(a+b*x)],x]},
  -1/(b*c*d^(n-1))*Subst[Int[SubstFor[(1-d^2*x^2)^((n-1)/2)/x^n,Cos[c*(a+b*x)]/d,u,x],x],x,Cos[c*(a+b*x)]/d] /;
 FunctionOfQ[Cos[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Tan] || EqQ[F,tan])
```

```
Int[u_*Coth[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Sinh[c*(a+b*x)],x]},
  1/(b*c*d^(n-1))*Subst[Int[SubstFor[(1+d^2*x^2)^((n-1)/2)/x^n,Sinh[c*(a+b*x)]/d,u,x],x],x,Sinh[c*(a+b*x)]/d] /;
 FunctionOfQ[Sinh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u]
```

```
Int[u_*Tanh[c_.*(a_.+b_.*x_)]^n_,x_Symbol] :=
  With[{d=FreeFactors[Cosh[c*(a+b*x)],x]},
  1/(b*c*d^(n-1))*Subst[Int[SubstFor[(-1+d^2*x^2)^((n-1)/2)/x^n,Cosh[c*(a+b*x)]/d,u,x],x],x,Cosh[c*(a+b*x)]/d] /;
 FunctionOfQ[Cosh[c*(a+b*x)]/d,u,x]] /;
FreeQ[{a,b,c},x] && IntegerQ[(n-1)/2] && NonsumQ[u]
```

6: $\int F[\text{Sin}[a+b\,x]]\,\left(v+d\,\text{Cos}[a+b\,x]^n\right)\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $\frac{n-1}{2}\in\mathbb{Z}$, then

$$\int F[\text{Sin}[a+b\,x]]\,\left(v+d\,\text{Cos}[a+b\,x]^n\right)\,dx \longrightarrow \int v\,F[\text{Sin}[a+b\,x]]\,dx+d\int F[\text{Sin}[a+b\,x]]\,\text{Cos}[a+b\,x]^n\,dx$$

Program code:

```
Int[u_*(v_+d_.*F_[c_.*(a_.+b_.*x_)]^n_.),x_Symbol] :=
  With[{e=FreeFactors[Sin[c*(a+b*x)],x]},
  Int[ActivateTrig[u*v],x] + d*Int[ActivateTrig[u]*Cos[c*(a+b*x)]^n,x] /;
 FunctionOfQ[Sin[c*(a+b*x)]/e,u,x]] /;
FreeQ[{a,b,c,d},x] && Not[FreeQ[v,x]] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Cos] || EqQ[F,cos])
```

```
Int[u_*(v_+d_.*F_[c_.*(a_.+b_.*x_)]^n_.),x_Symbol] :=
  With[{e=FreeFactors[Cos[c*(a+b*x)],x]},
  Int[ActivateTrig[u*v],x] + d*Int[ActivateTrig[u]*Sin[c*(a+b*x)]^n,x] /;
 FunctionOfQ[Cos[c*(a+b*x)]/e,u,x]] /;
FreeQ[{a,b,c,d},x] && Not[FreeQ[v,x]] && IntegerQ[(n-1)/2] && NonsumQ[u] && (EqQ[F,Sin] || EqQ[F,sin])
```

7: $\int F[\sin[a+b\,x]]\,\cos[a+b\,x]^n\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}$

Reference: G&R 2.503, CRC 483

Reference: G&R 2.502, CRC 482

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then $F[\sin[a+b\,x]]\,\cos[a+b\,x]^n = \frac{1}{b}\left(1-\sin[a+b\,x]^2\right)^{\frac{n-1}{2}} F[\sin[a+b\,x]]\,\partial_x \sin[a+b\,x]$

Rule: If $\frac{n-1}{2}\in\mathbb{Z}$, then

$$\int F[\sin[a+b\,x]]\,\cos[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\int \left(1-x^2\right)^{\frac{n-1}{2}} F[x]\,dx,\ x,\ \sin[a+b\,x]\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[F[Sin[a+b*x]]*Cos[a+b*x],x]","Subst[Int[F[x],x],x,Sin[a+b*x]]/b",Hold[
  With[{d=FreeFactors[Sin[v],x]},
  Dist[d/Coefficient[v,x,1],Subst[Int[SubstFor[1,Sin[v]/d,u/Cos[v],x],x],x,Sin[v]/d],x]]]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Sin[v],x],u/Cos[v],x]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  With[{d=FreeFactors[Sin[v],x]},
  Dist[d/Coefficient[v,x,1],Subst[Int[SubstFor[1,Sin[v]/d,u/Cos[v],x],x],x,Sin[v]/d],x]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Sin[v],x],u/Cos[v],x]]]
```

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[F[Cos[a+b*x]]*Sin[a+b*x],x]","-Subst[Int[F[x],x],x,Cos[a+b*x]]/b",Hold[
  With[{d=FreeFactors[Cos[v],x]},
  Dist[-d/Coefficient[v,x,1],Subst[Int[SubstFor[1,Cos[v]/d,u/Sin[v],x],x],x,Cos[v]/d],x]]]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Cos[v],x],u/Sin[v],x]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  With[{d=FreeFactors[Cos[v],x]},
  Dist[-d/Coefficient[v,x,1],Subst[Int[SubstFor[1,Cos[v]/d,u/Sin[v],x],x],x,Cos[v]/d],x]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Cos[v],x],u/Sin[v],x]]]
```

8. $\int u\left(a+b\,\text{Trig}[c+d\,x]^2+c\,\text{Trig}[c+d\,x]^2\right)^p\,dx$

1: $\int u\left(a+b\,\text{Cos}[c+d\,x]^2+c\,\text{Sin}[c+d\,x]^2\right)^p\,dx$ when $b-c=0$

Derivation: Algebraic simplification

Basis: If $b-c=0$, then $b\,\text{Cos}[z]^2+c\,\text{Sin}[z]^2=c$

Rule: If $b-c=0$, then

$$\int u\left(a+b\,\text{Tan}[d+e\,x]^2+c\,\text{Sec}[d+e\,x]^2\right)^p\,dx\;\longrightarrow\;(a+c)^p\int u\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*cos[d_.+e_.*x_]^2+c_.*sin[d_.+e_.*x_]^2)^p_.,x_Symbol] :=
  (a+c)^p*Int[ActivateTrig[u],x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[b-c,0]
```

2: $\int u\left(a+b\,\text{Tan}[c+d\,x]^2+c\,\text{Sec}[c+d\,x]^2\right)^p dx$ when $b+c==0$

Derivation: Algebraic simplification

Basis: If $b+c==0$, then $b\,\text{Tan}[z]^2+c\,\text{Sec}[z]^2==c$

Rule: If $b+c==0$, then

$$\int u\left(a+b\,\text{Tan}[d+e\,x]^2+c\,\text{Sec}[d+e\,x]^2\right)^p dx \longrightarrow (a+c)^p\int u\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*tan[d_.+e_.*x_]^2+c_.*sec[d_.+e_.*x_]^2)^p_.,x_Symbol] :=
  (a+c)^p*Int[ActivateTrig[u],x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[b+c,0]
```

```
Int[u_.*(a_.+b_.*cot[d_.+e_.*x_]^2+c_.*csc[d_.+e_.*x_]^2)^p_.,x_Symbol] :=
  (a+c)^p*Int[ActivateTrig[u],x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[b+c,0]
```

9. $\int y'[x]\, y[x]^m \, dx$

1: $\int \frac{y'[x]}{y[x]} \, dx$

Reference: G&R 2.111.1.2, CRC 27, A&S 3.3.15

Derivation: Integration by substitution and reciprocal rule for integration

Rule:

$$\int \frac{y'[x]}{y[x]} \, dx \longrightarrow \text{Log}[y[x]]$$

Program code:

```
Int[u_/y_,x_Symbol] :=
  With[{q=DerivativeDivides[ActivateTrig[y],ActivateTrig[u],x]},
    q*Log[RemoveContent[ActivateTrig[y],x]] /;
 Not[FalseQ[q]]] /;
Not[InertTrigFreeQ[u]]
```

```
Int[u_/(y_*w_),x_Symbol] :=
  With[{q=DerivativeDivides[ActivateTrig[y*w],ActivateTrig[u],x]},
    q*Log[RemoveContent[ActivateTrig[y*w],x]] /;
 Not[FalseQ[q]]] /;
Not[InertTrigFreeQ[u]]
```

2: $\int y'[x]\, y[x]^m \, dx$ when $m \neq -1$

Reference: G&R 2.111.1.1, CRC 23, A&S 3.3.14

Derivation: Integration by substitution and power rule for integration

Rule: If $m \neq -1$, then

$$\int y'[x]\, y[x]^m \, dx \longrightarrow \frac{y[x]^{m+1}}{m+1}$$

Program code:

```
Int[u_*y_^m_.,x_Symbol] :=
  With[{q=DerivativeDivides[ActivateTrig[y],ActivateTrig[u],x]},
   q*ActivateTrig[y^(m+1)]/(m+1) /;
 Not[FalseQ[q]]] /;
FreeQ[m,x] && NeQ[m,-1] && Not[InertTrigFreeQ[u]]
```

```
Int[u_*y_^m_.*z_^n_.,x_Symbol] :=
  With[{q=DerivativeDivides[ActivateTrig[y*z],ActivateTrig[u*z^(n-m)],x]},
   q*ActivateTrig[y^(m+1)*z^(m+1)]/(m+1) /;
 Not[FalseQ[q]]] /;
FreeQ[{m,n},x] && NeQ[m,-1] && Not[InertTrigFreeQ[u]]
```

10. $\int u\,(a\,F[c+d\,x]^p)^n\,dx$ when $F\in\{\text{Sin},\text{Cos},\text{Tan},\text{Cot},\text{Sec},\text{Csc}\}\land n\notin\mathbb{Z}\land p\in\mathbb{Z}$

1: $\int u\,(a\,F[c+d\,x]^p)^n\,dx$ when $F\in\{\text{Sin},\text{Cos},\text{Tan},\text{Cot},\text{Sec},\text{Csc}\}\land n\notin\mathbb{Z}\land p\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a\,F[c+d\,x]^p)^n}{F[c+d\,x]^{n\,p}} == 0$

Rule: If $F\in\{\text{Sin},\text{Cos},\text{Tan},\text{Cot},\text{Sec},\text{Csc}\}\land n\notin\mathbb{Z}\land p\in\mathbb{Z}$, then

$$\int u\,(a\,F[c+d\,x]^p)^n\,dx \longrightarrow \frac{(a\,F[c+d\,x]^p)^n}{F[c+d\,x]^{n\,p}}\int u\,F[c+d\,x]^{n\,p}\,dx$$

Program code:

```
Int[u_.*(a_.*F_[c_.+d_.*x_]^p_)^n_,x_Symbol] :=
  With[{v=ActivateTrig[F[c+d*x]]},
  a^IntPart[n]*(v/NonfreeFactors[v,x])^(p*IntPart[n])*(a*v^p)^FracPart[n]/NonfreeFactors[v,x]^(p*FracPart[n])*
    Int[ActivateTrig[u]*NonfreeFactors[v,x]^(n*p),x]] /;
FreeQ[{a,c,d,n,p},x] && InertTrigQ[F] && Not[IntegerQ[n]] && IntegerQ[p]
```

2: $\int u\,\left(a\,\left(b\,F[c+d\,x]\right)^p\right)^n\,dx$ when $F\in\{\text{Sin},\,\text{Cos},\,\text{Tan},\,\text{Cot},\,\text{Sec},\,\text{Csc}\}\,\land\,n\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$

- Derivation: Piecewise constant extraction

- Basis: $\partial_x\,\frac{\left(a\,\left(b\,F[c+d\,x]\right)^p\right)^n}{\left(b\,F[c+d\,x]\right)^{n\,p}}==0$

- Rule: If $F\in\{\text{Sin},\,\text{Cos},\,\text{Tan},\,\text{Cot},\,\text{Sec},\,\text{Csc}\}\,\land\,n\notin\mathbb{Z}\,\land\,p\notin\mathbb{Z}$, then

$$\int u\,\left(a\,\left(b\,F[c+d\,x]\right)^p\right)^n\,dx\,\longrightarrow\,\frac{a^{\text{IntPart}[n]}\,\left(a\,\left(b\,F[c+d\,x]\right)^p\right)^{\text{FracPart}[n]}}{\left(b\,F[c+d\,x]\right)^{p\,\text{FracPart}[n]}}\int u\,\left(b\,F[c+d\,x]\right)^{n\,p}\,dx$$

- Program code:

```
Int[u_.*(a_.*(b_.*F_[c_.+d_.*x_])^p_)^n_.,x_Symbol] :=
  With[{v=ActivateTrig[F[c+d*x]]},
  a^IntPart[n]*(a*(b*v)^p)^FracPart[n]/(b*v)^(p*FracPart[n])*Int[ActivateTrig[u]*(b*v)^(n*p),x]] /;
FreeQ[{a,b,c,d,n,p},x] && InertTrigQ[F] && Not[IntegerQ[n]] && Not[IntegerQ[p]]
```

11: $\int F[\tan[a+b\,x]]\,dx$ when $F[\tan[a+b\,x]]$ is free of inverse functions

Reference: G&R 2.504

Derivation: Integration by substitution

Basis: $F[\tan[z]] == \frac{F[\tan[z]]}{1+\tan[z]^2}\,\partial_z \tan[z]$

Rule: If $F[\tan[a+b\,x]]$ is free of inverse functions, then

$$\int F[\tan[a+b\,x]]\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int \frac{F[x]}{1+x^2}\,dx,\ x,\ \tan[a+b\,x]\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[F[Tan[a+b*x]],x]","1/b*Subst[Int[F[x]/(1+x^2),x],x,Tan[a+b*x]]",Hold[
  With[{d=FreeFactors[Tan[v],x]},
  Dist[d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v]/d,u,x],x],x,Tan[v]/d],x]]]]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Tan[v],x],u,x]] /;
SimplifyFlag && InverseFunctionFreeQ[u,x] &&
  Not[MatchQ[u,v_.*(c_.*tan[w_]^n_.*tan[z_]^n_.)^p_. /; FreeQ[{c,p},x] && IntegerQ[n] && LinearQ[w,x] && EqQ[z,2*w]]],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  With[{d=FreeFactors[Tan[v],x]},
  Dist[d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v]/d,u,x],x],x,Tan[v]/d],x]] /;
 Not[FalseQ[v]] && FunctionOfQ[NonfreeFactors[Tan[v],x],u,x]] /;
InverseFunctionFreeQ[u,x] &&
  Not[MatchQ[u,v_.*(c_.*tan[w_]^n_.*tan[z_]^n_.)^p_. /; FreeQ[{c,p},x] && IntegerQ[n] && LinearQ[w,x] && EqQ[z,2*w]]]]
```

12: $\int u\,(c\,\text{Sin}[v])^m\,dx$ when $v == a+b\,x\,\land\,m+\frac{1}{2}\in\mathbb{Z}\,\land\,u\,\text{Sin}\left[\frac{v}{2}\right]^{2m}$ is a function of $\text{Tan}\left[\frac{v}{2}\right]$ free of inverse functions

Derivation: Piecewise constant extraction

Basis: If $v == a+b\,x$, then $\partial_x\frac{(c\,\text{Sin}[v])^m\,\left(c\,\text{Tan}\left[\frac{v}{2}\right]\right)^m}{\text{Sin}\left[\frac{v}{2}\right]^{2m}} == 0$

Rule: If $v == a+b\,x\,\land\,m+\frac{1}{2}\in\mathbb{Z}\,\land\,u\,\text{Sin}\left[\frac{v}{2}\right]^{2m}$ is a function of $\text{Tan}\left[\frac{v}{2}\right]$ free of inverse functions, then

$$\int u\,(c\,\text{Sin}[v])^m\,dx\,\longrightarrow\,\frac{(c\,\text{Sin}[v])^m\,\left(c\,\text{Tan}\left[\frac{v}{2}\right]\right)^m}{\text{Sin}\left[\frac{v}{2}\right]^{2m}}\int\frac{u\,\text{Sin}\left[\frac{v}{2}\right]^{2m}}{\left(c\,\text{Tan}\left[\frac{v}{2}\right]\right)^m}\,dx$$

Program code:

```
Int[u_*(c_.*sin[v_])^m_,x_Symbol] :=
  With[{w=FunctionOfTrig[u*Sin[v/2]^(2*m)/(c*Tan[v/2])^m,x]},
   (c*Sin[v])^m*(c*Tan[v/2])^m/Sin[v/2]^(2*m)*Int[u*Sin[v/2]^(2*m)/(c*Tan[v/2])^m,x] /;
 Not[FalseQ[w]] && FunctionOfQ[NonfreeFactors[Tan[w],x],u*Sin[v/2]^(2*m)/(c*Tan[v/2])^m,x]] /;
FreeQ[c,x] && LinearQ[v,x] && IntegerQ[m+1/2] && Not[SumQ[u]] && InverseFunctionFreeQ[u,x]
```

13: $\int u\,\left(a\,\text{Tan}[c+d\,x]^n+b\,\text{Sec}[c+d\,x]^n\right)^p\,dx$ when $(n\mid p)\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $n\in\mathbb{Z}$, then $a\,\text{Tan}[z]^n+b\,\text{Sec}[z]^n=\text{Sec}[z]^n\,\left(b+a\,\text{Sin}[z]^n\right)$

Rule: If $(n\mid p)\in\mathbb{Z}$, then

$$\int u\,\left(a\,\text{Tan}[c+d\,x]^n+b\,\text{Sec}[c+d\,x]^n\right)^p\,dx\;\longrightarrow\;\int u\,\text{Sec}[c+d\,x]^{n\,p}\,\left(b+a\,\text{Sin}[c+d\,x]^n\right)^p\,dx$$

Program code:

```
Int[u_.*(a_.*tan[c_.+d_.*x_]^n_.+b_.*sec[c_.+d_.*x_]^n_.)^p_,x_Symbol] :=
  Int[ActivateTrig[u]*Sec[c+d*x]^(n*p)*(b+a*Sin[c+d*x]^n)^p,x] /;
FreeQ[{a,b,c,d},x] && IntegersQ[n,p]
```

```
Int[u_.*(a_.*cot[c_.+d_.*x_]^n_.+b_.*csc[c_.+d_.*x_]^n_.)^p_,x_Symbol] :=
  Int[ActivateTrig[u]*Csc[c+d*x]^(n*p)*(b+a*Cos[c+d*x]^n)^p,x] /;
FreeQ[{a,b,c,d},x] && IntegersQ[n,p]
```

14. $\int u\,\left(a\,\text{Trig}[c+d\,x]^p+b\,\text{Trig}[c+d\,x]^q+\cdots\right)^n\,dx$

1: $\int u\,\left(a\,\text{Trig}[c+d\,x]^p+b\,\text{Trig}[c+d\,x]^q\right)^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: $a\,z^p+b\,z^q==z^p\,\left(a+b\,z^{q-p}\right)$

Rule: If $n\in\mathbb{Z}$, then

$$\int u\,\left(a\,\text{Trig}[c+d\,x]^p+b\,\text{Trig}[c+d\,x]^q\right)^n\,dx\;\longrightarrow\;\int u\,\text{Trig}[c+d\,x]^{n\,p}\,\left(a+b\,\text{Trig}[c+d\,x]^{q-p}\right)^n\,dx$$

Program code:

```
Int[u_*(a_*F_[c_.+d_.*x_]^p_.+b_.*F_[c_.+d_.*x_]^q_.)^n_.,x_Symbol] :=
  Int[ActivateTrig[u*F[c+d*x]^(n*p)*(a+b*F[c+d*x]^(q-p))^n],x] /;
FreeQ[{a,b,c,d,p,q},x] && InertTrigQ[F] && IntegerQ[n] && PosQ[q-p]
```

2: $\int u\,\left(a\,\text{Trig}[d+e\,x]^p+b\,\text{Trig}[d+e\,x]^q+c\,\text{Trig}[d+e\,x]^r\right)^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: $a\,z^p+b\,z^q+c\,z^r == z^p\,\left(a+b\,z^{q-p}+c\,z^{r-p}\right)$

Rule: If $n\in\mathbb{Z}$, then

$$\int u\,\left(a\,\text{Trig}[d+e\,x]^p+b\,\text{Trig}[d+e\,x]^q+c\,\text{Trig}[d+e\,x]^r\right)^n\,dx \longrightarrow \int u\,\text{Trig}[d+e\,x]^{n\,p}\,\left(a+b\,\text{Trig}[d+e\,x]^{q-p}+c\,\text{Trig}[d+e\,x]^{r-p}\right)^n\,dx$$

Program code:

```
Int[u_*(a_*F_[d_.+e_.*x_]^p_.+b_.*F_[d_.+e_.*x_]^q_.+c_.*F_[d_.+e_.*x_]^r_.)^n_.,x_Symbol] :=
  Int[ActivateTrig[u*F[d+e*x]^(n*p)*(a+b*F[d+e*x]^(q-p)+c*F[d+e*x]^(r-p))^n],x] /;
FreeQ[{a,b,c,d,e,p,q,r},x] && InertTrigQ[F] && IntegerQ[n] && PosQ[q-p] && PosQ[r-p]
```

15: $\int u\,\left(a+b\,\text{Trig}[d+e\,x]^p+c\,\text{Trig}[d+e\,x]^{-p}\right)^n\,dx$ when $n\in\mathbb{Z}\land p<0$

Derivation: Algebraic simplification

Basis: $a+b\,z^p+c\,z^q == z^p\,\left(b+a\,z^{-p}+c\,z^{q-p}\right)$

Rule: If $n\in\mathbb{Z}\land p<0$, then

$$\int u\,\left(a+b\,\text{Trig}[d+e\,x]^p+c\,\text{Trig}[d+e\,x]^q\right)^n\,dx \longrightarrow \int u\,\text{Trig}[d+e\,x]^{n\,p}\,\left(b+a\,\text{Trig}[d+e\,x]^{-p}+c\,\text{Trig}[d+e\,x]^{q-p}\right)^n\,dx$$

Program code:

```
Int[u_*(a_+b_.*F_[d_.+e_.*x_]^p_.+c_.*F_[d_.+e_.*x_]^q_.)^n_.,x_Symbol] :=
  Int[ActivateTrig[u*F[d+e*x]^(n*p)*(b+a*F[d+e*x]^(-p)+c*F[d+e*x]^(q-p))^n],x] /;
FreeQ[{a,b,c,d,e,p,q},x] && InertTrigQ[F] && IntegerQ[n] && NegQ[p]
```

16: $\int u\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx$ when $a^2+b^2==0$

Derivation: Algebraic simplification

Basis: If $a^2+b^2==0$, then $a\,\mathrm{Cos}[z]+b\,\mathrm{Sin}[z]==a\,e^{-\frac{a\,z}{b}}$

Rule: If $a^2+b^2==0$, then

$$\int u\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx\;\longrightarrow\;\int u\left(a\,e^{-\frac{a\,(c+d\,x)}{b}}\right)^n\,dx$$

Program code:

```
Int[u_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  Int[ActivateTrig[u]*(a*E^(-a/b*(c+d*x)))^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2+b^2,0]
```

17: $\int u\,dx$ when TrigSimplifyQ[u]

Rule: If TrigSimplifyQ[u], then

$$\int u\,dx\;\longrightarrow\;\int \mathrm{TrigSimplify}[u]\,dx$$

Program code:

```
Int[u_,x_Symbol] :=
  Int[TrigSimplify[u],x] /;
TrigSimplifyQ[u]
```

18. $\int u\,(v^m\,w^n\,\cdots)^p\,dx$ when $p\notin\mathbb{Z}$

1: $\int u\,(a\,v)^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a\,F[x])^p}{F[x]^p} == 0$

Rule: If $p\notin\mathbb{Z}$, then

$$\int u\,(a\,v)^p\,dx \longrightarrow \frac{a^{\text{IntPart}[p]}\,(a\,v)^{\text{FracPart}[p]}}{v^{\text{FracPart}[p]}}\int u\,v^p\,dx$$

–

Program code:

```
Int[u_.*(a_*v_)^p_,x_Symbol] :=
  With[{uu=ActivateTrig[u],vv=ActivateTrig[v]},
  a^IntPart[p]*(a*vv)^FracPart[p]/(vv^FracPart[p])*Int[uu*vv^p,x]] /;
FreeQ[{a,p},x] && Not[IntegerQ[p]] && Not[InertTrigFreeQ[v]]
```

2: $\int u\,(v^m)^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(F[x]^m)^p}{F[x]^{m\,p}} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int u\,(v^m)^p\,dx \longrightarrow \frac{(v^m)^{\text{FracPart}[p]}}{v^{m\,\text{FracPart}[p]}} \int u\,v^{m\,p}\,dx$$

Program code:

```
Int[u_.*(v_^m_)^p_,x_Symbol] :=
  With[{uu=ActivateTrig[u],vv=ActivateTrig[v]},
  (vv^m)^FracPart[p]/(vv^(m*FracPart[p]))*Int[uu*vv^(m*p),x]] /;
FreeQ[{m,p},x] && Not[IntegerQ[p]] && Not[InertTrigFreeQ[v]]
```

3: $\int u\,(v^m w^n)^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(F[x]^m G[x]^n)^p}{F[x]^{m p} G[x]^{n p}} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int u\,(v^m w^n)^p\,dx \longrightarrow \frac{(v^m w^n)^{\text{FracPart}[p]}}{v^{m\,\text{FracPart}[p]}\,w^{n\,\text{FracPart}[p]}}\int u\,v^{m p}\,w^{n p}\,dx$$

Program code:

```
Int[u_.*(v_^m_.*w_^n_.)^p_,x_Symbol] :=
  With[{uu=ActivateTrig[u],vv=ActivateTrig[v],ww=ActivateTrig[w]},
  (vv^m*ww^n)^FracPart[p]/(vv^(m*FracPart[p])*ww^(n*FracPart[p]))*Int[uu*vv^(m*p)*ww^(n*p),x]] /;
FreeQ[{m,n,p},x] && Not[IntegerQ[p]] && (Not[InertTrigFreeQ[v]] || Not[InertTrigFreeQ[w]])
```

19: $\int u\,dx$ when `ExpandTrig[u, x]` is a sum

Derivation: Algebraic expansion

–

Rule: If `ExpandTrig[u, x]` is a sum, then

$$\int u\,dx \longrightarrow \int \text{ExpandTrig}[u,\,x]\,dx$$

–

Program code:

```
Int[u_,x_Symbol] :=
  With[{v=ExpandTrig[u,x]},
  Int[v,x] /;
 SumQ[v]] /;
Not[InertTrigFreeQ[u]]
```

20: $\int F[\sin[a+b x], \cos[a+b x]]\,dx$ when $F[\sin[a+b x], \cos[a+b x]]$ is free of inverse functions and $\int \frac{1}{1+x^2} F\left[\frac{2x}{1+x^2}, \frac{1-x^2}{1+x^2}\right] dx$ is integrable in closed –form

Reference: G&R 2.501, CRC 484

Derivation: Integration by substitution

Basis: $F[\sin[a+b x], \cos[a+b x]] == \frac{2}{b}\,\text{Subst}\left[\frac{1}{1+x^2} F\left[\frac{2x}{1+x^2}, \frac{1-x^2}{1+x^2}\right], x, \tan\left[\frac{a+b x}{2}\right]\right] \partial_x \tan\left[\frac{a+b x}{2}\right]$

Rule: If $F[\sin[a+b x], \cos[a+b x]]$ is free of inverse functions and $\int \frac{1}{1+x^2} F\left[\frac{2x}{1+x^2}, \frac{1-x^2}{1+x^2}\right] dx$ is integrable in closed-form, then

$$\int F[\sin[a+b x], \cos[a+b x]]\,dx \to \frac{2}{b}\,\text{Subst}\left[\int \frac{1}{1+x^2} F\left[\frac{2x}{1+x^2}, \frac{1-x^2}{1+x^2}\right] dx, x, \tan\left[\frac{a+b x}{2}\right]\right]$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{w=Block[{$ShowSteps=False,$StepCounter=Null},
            Int[SubstFor[1/(1+FreeFactors[Tan[FunctionOfTrig[u,x]/2],x]^2*x^2),Tan[FunctionOfTrig[u,x]/2]/FreeFactors[Tan[FunctionOfTrig[u,x]/2]
  ShowStep["","Int[F[Sin[a+b*x],Cos[a+b*x]],x]","2/b*Subst[Int[1/(1+x^2)*F[2*x/(1+x^2),(1-x^2)/(1+x^2)],x],x,Tan[(a+b*x)/2]]",Hold[
  Module[{v=FunctionOfTrig[u,x],d},
  d=FreeFactors[Tan[v/2],x];
  Dist[2*d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v/2]/d,u,x],x],x,Tan[v/2]/d],x]]]] /;
 CalculusFreeQ[w,x]] /;
SimplifyFlag && InverseFunctionFreeQ[u,x] && Not[FalseQ[FunctionOfTrig[u,x]]],

Int[u_,x_Symbol] :=
  With[{w=Block[{$ShowSteps=False,$StepCounter=Null},
            Int[SubstFor[1/(1+FreeFactors[Tan[FunctionOfTrig[u,x]/2],x]^2*x^2),Tan[FunctionOfTrig[u,x]/2]/FreeFactors[Tan[FunctionOfTrig[u,x]/2]
  Module[{v=FunctionOfTrig[u,x],d},
  d=FreeFactors[Tan[v/2],x];
  Dist[2*d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v/2]/d,u,x],x],x,Tan[v/2]/d],x]] /;
 CalculusFreeQ[w,x]] /;
InverseFunctionFreeQ[u,x] && Not[FalseQ[FunctionOfTrig[u,x]]]]
```

```
(* If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[F[Sin[a+b*x],Cos[a+b*x]],x]","2/b*Subst[Int[1/(1+x^2)*F[2*x/(1+x^2),(1-x^2)/(1+x^2)],x],x,Tan[(a+b*x)/2]]",Hold[
  With[{d=FreeFactors[Tan[v/2],x]},
  Dist[2*d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v/2]/d,u,x],x],x,Tan[v/2]/d],x]]]] /;
 Not[FalseQ[v]]] /;
SimplifyFlag && InverseFunctionFreeQ[u,x],

Int[u_,x_Symbol] :=
  With[{v=FunctionOfTrig[u,x]},
  With[{d=FreeFactors[Tan[v/2],x]},
  Dist[2*d/Coefficient[v,x,1],Subst[Int[SubstFor[1/(1+d^2*x^2),Tan[v/2]/d,u,x],x],x,Tan[v/2]/d],x]] /;
 Not[FalseQ[v]]] /;
InverseFunctionFreeQ[u,x]] *)
```

X: $\int F[\text{Trig}[a+b\,x]]\,dx$

- Note: If integrand involves inert trig functions, must suppress further application of integration rules.

- Rule:

$$\int F[\text{Trig}[a+b\,x]]\,dx \longrightarrow \int F[\text{Trig}[a+b\,x]]\,dx$$

- Program code:

```
Int[u_,x_Symbol] :=
  With[{v=ActivateTrig[u]},
   CannotIntegrate[v,x]] /;
Not[InertTrigFreeQ[u]]
```